

The Fundamental Theorem of Calculus

> **restart;**

The Fundamental Theorem, Part I

The first part of the Fundamental Theorem states that for any continuous function f on $[a, b]$ the

function $F(x) = \int_a^x f(t) dt$ is differentiable on $[a, b]$ and

$$\frac{d}{dx} \int_a^x f(t) dt = f(x) .$$

In other words, if we have a **definite integral with variable upper limit, then the rate of change with respect to the upper limit is precisely the value of the integrand evaluated at the upper limit** .

The following procedure produces an animation which illustrates part I of the Fundamental Theorem.

As input it takes

a function f ,

an interval, specified by a and b ,

a parameter n determining the number of frames used for the animation,

a parameter $withF$. If $withF = 1$ then the function F is also included in the plot (otherwise set $withF = 0$).

The information printed during the animation is

the current area under the graph, i.e. $F(x) = \int_a^x f(t) dt$, for the current value of x ,

the area under the graph in the previous frame,

the approximate rate of change of the area, i.e. (current area - previous area)/(change in x), or

$$\frac{F(x_i) - F(x_{i-1})}{\Delta x_i}$$

the exact rate of change given by $f(x)$,

the exact rate of change given by $\frac{d}{dx} F(x)$.

```

> FT_graph := proc(f, a, b, n, withF)
local dF, F, i, j, deltaF, M, m, cp, ip, ap, tp1, tp2, tp3, tp4, tp5, textmargin, T, H;
F := unapply(int(f(t), t=a..x), x);
dF := D(F);
M := max(0, evalf(maximize(f(x), x, a..b)));
m := min(0, evalf(minimize(f(x), x, a..b)));
textmargin := (M-m)*.2;
T := textmargin/5;
H := M + textmargin;
m := m - (M-m)*.05;
x.0 := a;
for i from 1 to n do
x.i := a+(i-1)/(n-1)*(b-a);
F.i := F(x.i);
if i > 1 then
deltaF := (F.i-F.(i-1))/(x.i-x.(i-1));
else
deltaF := 0;
fi;
s.i:=[[x.i,0], [x.(i-1),0], seq([x.(i-1)+j*(x.i-x.(i-1))/5,f(x.(i-1)+j*(x.i-x.(i-1))/5)], j=0..5)];
cp := plot(f(t), t=a..x.i, color=red, labels=[t,'f']);
if withF = 1 then
ip := plot(F(x), x=a..x.i, color=blue);
fi;
ap:=plots[polygonplot]({seq(s.j, j=1..i)}, color=cyan, view=[a..b,m..H], style=patchnograd);
tp1:=plots[textplot]([(a+b)/2,H,cat("current area: ",convert(evalf(F.i,5),string))], color=black);
if i > 1 then
tp2:=plots[textplot]([(a+b)/2,M+3.75*T,cat("previous area: ",convert(evalf(F.(i-1),5),string))],
color=black);
tp3:=plots[textplot]([(a+b)/2,M+2.5*T,cat("approx. rate of change: ",convert(evalf
(deltaF,5),string))], color=black);
tp4:=plots[textplot]([(a+b)/2,M+1.25*T,cat("f(x): ",convert(evalf(f(x.i),5),string))], color=black);
tp5:=plots[textplot]([(a+b)/2,M,cat("D(F)(x): ",convert(evalf(dF(x.i),5),string))], color=black);
else
tp2:=plots[textplot]([(a+b)/2,M+3.75*T,"previous area: "], color=black);
tp3:=plots[textplot]([(a+b)/2,M+2.5*T,"approx. rate of change: "], color=black);
tp4:=plots[textplot]([(a+b)/2,M+1.25*T,"f(x): "], color=black);

```

```

tp5:=plots[textplot]([(a+b)/2,M,"D(F)(x): "], color=black):
fi;
if withF = 1 then
p.i := plots[display](cp, ip, ap, tp1, tp2, tp3, tp4, tp5):
else
p.i := plots[display](cp, ap, tp1, tp2, tp3, tp4, tp5):
fi;
od:
plots[display](seq(p.i, i=1..n), insequence=true);
end:

```

Here is a sample call to **FT_graph** for the function $F(x) = \int_{-1}^x \sqrt{|t|} - \frac{3t}{2} + 1 dt$ and x in the

interval $[-1, 3]$.


Run the animation and observe what is going on.

```

> f:=x->sqrt(abs(x))-3*x/2+1;
a:=-1:
b:=3:
n:=20:
FT_graph(f,a,b,n,0);

```

$$f := x \rightarrow \sqrt{|x|} - \frac{3}{2}x + 1$$



The Fundamental Theorem, Part II

The second part of the Fundamental Theorem states that for every continuous function f on $[a, b]$

$$\int_a^b f(x) dx = F(b) - F(a) \quad ,$$

where F is some antiderivative of f .

Thus, this gives **a way to evaluate definite integrals**.

We use the same function f as above and compute an antiderivative F .

**> f:=x->sqrt(abs(x))-3*x/2+1;
F:=unapply(int(f(x), x),x);**

$$f := x \rightarrow \sqrt{|x|} - \frac{3}{2}x + 1$$

$$F := x \rightarrow \frac{2}{3}\sqrt{|x|}x - \frac{3}{4}x^2 + x$$

Let's evaluate the integral $\int_{-1}^3 f(x) dx$ by using Maple's **int** command as a black box, and by using the

second part of the Fundamental Theorem, i.e., by evaluating F at the limits of the definite integral.

**> Int(f(x), x=-1..3) = int(f(x), x=-1..3);
'F(3)-F(-1)'=F(3)-F(-1);**


$$\int_{-1}^3 \sqrt{|x|} - \frac{3}{2}x + 1 dx = 2\sqrt{3} - \frac{4}{3}$$

$$F(3) - F(-1) = 2\sqrt{3} - \frac{4}{3}$$

Finally, we plot f and F together over the interval $[-1, 3]$.

Note that F here is not quite the same as the F in example illustrating the first part of the Fundamental Theorem.

> `plot([f(x), F(x)], x=-1..3, color=[red,green]);`



Assignment 11

Ex.1:

How can the difference between the F 's in the illustration of the the two parts of the Fundamental Theorem above be explained?

Ex.2:

Use the procedure **FT_graph** to study the statement of the Fundamental Theorem, part I, for the function $f(x) = 2x^2 - x^3$ on the interval $[-1, 3]$.

Run the animation step by step (using the `->` button) and describe in your own words what is happening.

Include the graph of F in the animation (and your discussion).

Ex.3:

Let $f(x) = x^3 - 4x^2 + 3x$, and $[a, b] = [0, 4]$.

Also let $F(x) = \int_a^x f(t) dt$, as in part I of the Fundamental Theorem.

a) Use **FT_graph** to plot f and F together over $[a, b]$.

b) Solve $D(F)(x) = 0$. What can you see to be true about the graphs of f and F at points where $D(F)(x) = 0$?

c) Over what intervals is the function F increasing/decreasing?

What is true about f over those intervals?

d) Calculate the derivative $D(f)$ and plot it together with F .

e) Solve $D(f)(x) = 0$. What can you see to be true about the graph of F at points where $D(f)(x) = 0$?

Ex.4:

The **Fresnel function** is defined as $S(x) = \int_0^x \sin\left(\frac{\pi t^2}{2}\right) dt$.

a) At what values of x does this function have local maximum values?

b) Find the coordinates of the first inflection point to the right of the origin.

c) Does the Fresnel function have a horizontal asymptote?

d) Plot $\sin\left(\frac{\pi x^2}{2}\right)$ and S together on $[0, 10]$.

e) Use the graph of the Fresnel function to solve the equation $S(x) = .2$ correct to one decimal place.

f) Use Maple's **fsolve** command to verify your answer in e).

Ex.5:

A high-tech company purchases a new computer system whose initial value is V .

The system will depreciate at the rate $f = f(t)$ and will accumulate maintenance costs at the rate $g = g(t)$, where t is the time measured in months.

The company wants to determine the optimal time to replace the system.

a) Why will the company choose to replace the system at the time T for which the function

$$C(t) = \frac{1}{t} \int_0^t f(s) + g(s) ds$$

attains its minimum? What does C represent?

b) Suppose that $f(t) = \frac{V}{15} - \frac{Vt}{450}$ for t between 0 and 30, and $f(t) = 0$ for $30 < t$.

Determine the length of time T for the total depreciation $D(t) = \int_0^t f(s) ds$ to equal the initial value V .

Hint: We define f as a piecewise function. The resulting integral can be converted back to a piecewise function using the [convert,piecewise](#) command (click for help).

```
> f := t->piecewise(t<=30, V/15-V*t/450,
t>30, 0);
f(t);
```

$$f := t \rightarrow \text{piecewise}\left(t \leq 30, \frac{1}{15}V - \frac{1}{450}Vt, 30 < t, 0\right)$$

$$C'(t) = \begin{cases} \frac{1}{15}V - \frac{1}{450}Vt & t \leq 30 \\ 0 & 30 < t \end{cases}$$

c) Further, assume that $g(t) = \frac{Vt^2}{12900}$ for $0 < t \leq T$.

It can be shown that the critical numbers of C occur when $C'(t) = f'(t) + g'(t) = 0$.

Determine the absolute minimum of C on $[0, T]$.

d) Graph C and $f+g$ together in one plot over $[0, T]$ in order to verify the claim about the critical points of C made above. (Hint: pick a suitable value of C for the plot.)

e) Create another plot showing the total depreciation, total maintenance cost and C together over $[0, T]$. Comment.